

```
1 ' -----
2 '     From the book Inside Active Directory, ISBN 0-201-61621-1
3 '     Copyright (C) 2002 by Addison-Wesley
4 '     Script by Sakari Kouti (see http://www.kouti.com)
5 ' You have a royalty-free right to use, modify, reproduce and distribute
6 ' this script (and/or any modified version) in any way you find useful,
7 ' provided that you agree that Addison-Wesley or Sakari Kouti has no
8 ' warranty, obligations or liability for the script. If you modify
9 ' the script, you must retain this copyright notice.
10 ' -----
11 Option Explicit
12
13 Dim objExcel, objClass, strPropName, strGivenClass, strSheetName
14
15 strGivenClass = InputBox("Enter a class LDAP name")
16
17 Set objExcel = WScript.CreateObject("Excel.Application")
18 objExcel.Visible = True
19 objExcel.Workbooks.Add
20
21 strSheetName = "Attrs of " & strGivenClass
22 If Len(strSheetName) > 31 Then _
23     strSheetName = Left(strSheetName,28) & "..."
24 objExcel.ActiveSheet.Name = strSheetName
25 objExcel.ActiveSheet.Range("A1").Activate
26 objExcel.ActiveCell.Value = "Attr LDAP Name"
27 objExcel.ActiveCell.Offset(0, 1).Value = "M/O"
28 objExcel.ActiveCell.Offset(0, 2).Value = "Syntax"
29 objExcel.ActiveCell.Offset(0, 3).Value = "MultiValued"
30 objExcel.ActiveCell.Offset(0, 4).Value = "MinRange"
31 objExcel.ActiveCell.Offset(0, 5).Value = "MaxRange"
32 objExcel.ActiveCell.Offset(0, 6).Value = "OID"
33 objExcel.ActiveCell.Offset(1, 0).Activate
34
35 Set objClass = GetObject("LDAP://schema/" & strGivenClass)
36
37 For Each strPropName In objClass.MandatoryProperties
38     Call ShowAttrAndFeatures("Mandatory", strPropName)
39 Next
40
41 For Each strPropName In objClass.OptionalProperties
42     Call ShowAttrAndFeatures("Optional", strPropName)
43 Next
44
45 Sub ShowAttrAndFeatures(strManOrOpt, strPropName)
46     Dim objAttribute
47
48     On Error Resume Next
49
50     Set objAttribute = GetObject("LDAP://schema/" & strPropName)
51
52     objExcel.ActiveCell.Value = strPropName
53     objExcel.ActiveCell.Offset(0, 1).Value = strManOrOpt
54     With objAttribute
55         objExcel.ActiveCell.Offset(0, 2).Value = .Syntax
56         objExcel.ActiveCell.Offset(0, 3).Value = .MultiValued
57         objExcel.ActiveCell.Offset(0, 4).Value = .MinRange
58         objExcel.ActiveCell.Offset(0, 5).Value = .MaxRange
59         objExcel.ActiveCell.Offset(0, 6).Value = .OID
60     End With
61     objExcel.ActiveCell.Offset(1, 0).Activate
62 End Sub
63
```